

# How to Color Your Report?

By Xuefeng Yu, Celgene Co., Summit, NJ

## ABSTRACT

A colorful report with highlighted background or bright color numbers will greatly facilitate the reviewer to identify the flags, the special values, conditions, or outliers. SAS® has provided various ways to make it an easy task. This article summarizes the different methods to produce a colorful report, for example, by using PROC FORMAT, PROC REPORT, or using ODS Style definitions. SAS® codes and data examples and the result outputs are given to illustrate the programming ideas.

## INTRODUCTION

This paper presents a few different ways to construct a colorful PDF report using ODS, PROC REPORT, PROC FORMAT. Depending upon the data and how it is organized, as well as the table designed, one way may be easier or more convenient than the other. Some methods have been discussed in the referenced papers. Here I try to use simplified examples to illustrate step by step on how to build colorful tables.

## METHOD 1: ODS STYLE / PROC TEMPLATE

SAS® has provided a powerful tool, ODS, to enable users output their report to RTF, PDF or HTML files. Although HTML files will require more attention in the future, we will still focus on PDF files in this paper for our current needs. Simple codes below allow users to output the report with SAS® default STYLE.

```
ODS LISTING CLOSE;  
ODS PDF file='file name';  
PROC REPORT .....;  
.....
```

```
ODS _ALL_ CLOSE;  
ODS LISTING;
```

Besides the default STYLE, SAS® also provides some pre-defined styles, such as SASWEB, STATISTICAL, and ANALYSIS etc. Users can specify a STYLE name other than default in ODS PDF statement, e.g.

```
ODS PDF file= 'filename' STYLE=SASWEB;
```

Users also have the option to create their own STYLES by redefining the pre-defined style elements and attributes to customize fonts, colors, and borders, etc. PROC TEMPLATE is the procedure for this purpose. Lauren Haworth [1] has explained how to use PROC TEMPLATE to modify STYLES and gave some examples. You may refer to that paper if interested. We will skip this part in this paper.

The advantage of using PROC TEMPLATE to customize STYLES is that it can be saved and easily shared among users once the new STYLE is established. It can change the color of the table header and its contents, and also the color of the font and the background. However, it's difficult to set up such a STYLE for some special outputs that we will see in the later examples.

## METHOD 2: PROC REPORT / CALL DEFINE

A shortcut way to highlight a table is to change the style in PROC REPORT procedure. You may change the other attributes besides background color as well by using COMPUTE statement and CALL DEFINE statement in PROC REPORT.

We will see how to change the background color, in other words "highlight", with a simplified lab data as below:

PID	Visit	Lab Test	Value	Unit	Range Low	Range High	Abnormal Flag
1001	BASELINE	ALBUMIN	39	g/L	32	50	N
1001	BASELINE	ALT	35	U/L	0	55	N
1001	BASELINE	ALKALINE PHOSPHATASE	154	U/L	37	147	H
1001	BASELINE	AST	33	U/L	0	45	N
1001	BASELINE	BICARBONATE	29	mmol/L	23	30	N
1001	BASELINE	BILIRUBIN, TOTAL	8.6	umol/L	5.1	25.7	N
1001	BASELINE	CREATININE	62	umol/L	62	115	N
1001	BASELINE	SODIUM	137	mmol/L	135	148	N
1001	CYCLE 1 DAY 1	ALBUMIN	41	g/L	32	50	N
1001	CYCLE 1 DAY 1	ALT	29	U/L	0	55	N
1001	CYCLE 1 DAY 1	ALKALINE PHOSPHATASE	170	U/L	37	147	H
1001	CYCLE 1 DAY 1	AST	33	U/L	0	45	N
1001	CYCLE 1 DAY 1	BICARBONATE	28	mmol/L	23	30	N
1001	CYCLE 1 DAY 1	BILIRUBIN, TOTAL	8.6	umol/L	5.1	25.7	N
1001	CYCLE 1 DAY 1	CREATININE	71	umol/L	62	115	N
1001	CYCLE 1 DAY 1	SODIUM	134	mmol/L	135	148	L
1001	CYCLE 2 DAY 1	ALBUMIN	38	g/L	32	50	N
1001	CYCLE 2 DAY 1	ALT	24	U/L	0	55	N
1001	CYCLE 2 DAY 1	ALKALINE PHOSPHATASE	131	U/L	37	147	N
1001	CYCLE 2 DAY 1	AST	21	U/L	0	45	N
1001	CYCLE 2 DAY 1	BICARBONATE	26	mmol/L	23	30	N
1001	CYCLE 2 DAY 1	BILIRUBIN, TOTAL	5.1	umol/L	5.1	25.7	N
1001	CYCLE 2 DAY 1	CREATININE	71	umol/L	62	115	N
1001	CYCLE 2 DAY 1	SODIUM	137	mmol/L	135	148	N

Suppose we want to report the lab data with a table. The rows with abnormal high values are highlighted in red, and the rows with abnormal low values are highlighted in blue. The column of lab test names are highlighted in 'LIGHTYELLOW', and the column of visit date is highlighted in 'MISTYROSE'. The code of PROC REPORT is listed below:

```
proc report data=db.clab nowd headline headskip spacing=1 split='~';
  column pt visname labdate lbseq labtest sival siunit silo sihi labflag ;
  define pt / order width=5 'PID' left;
  define visname / order width=14 'Visit' left;
  define labdate / order width=10 'Visit Date' center;
  define lbseq / order noprint;
  define labtest / display width=25 'Lab Test' left;
  define sival / display width=8 'Value' right;
  define siunit / display width=10 'Unit' center;
  define silo / display width=8 'Range~Low' right;
  define sihi / display width=8 'Range~High ' right;
  define labflag / display width=10 'Abnormal~Flag' center;

  COMPUTE pt;
  CALL DEFINE(_COL_, "style", "STYLE=[BACKGROUND=MISTYROSE]");
  ENDCOMP;
```

```

COMPUTE visname;
  CALL DEFINE(_COL_, "style", "STYLE=[BACKGROUND=MISTYROSE]");
ENDCOMP;
COMPUTE labdate;
  CALL DEFINE(_COL_, "style", "STYLE=[BACKGROUND=MISTYROSE]");
ENDCOMP;
COMPUTE labtest;
  CALL DEFINE(_COL_, "style", "STYLE=[BACKGROUND=LIGHTYELLOW]");
ENDCOMP;
COMPUTE labflag;
  IF labflag='H'
    THEN CALL DEFINE(_ROW_, "style", "STYLE=[BACKGROUND=RED]");
ENDCOMP;
COMPUTE silo;
  IF input(sival,best.) < silo
    THEN CALL DEFINE(_ROW_, "style", "STYLE=[BACKGROUND=BLUE]");
ENDCOMP;
run;

```

The keyword `_ROW_` indicates the entire row will be highlighted. Similarly, the keyword `_COL_` indicates the entire column will be highlighted. Please be noted that the `STYLE=` parameter can also be used to modify other attributes. For example, `STYLE=[font_size=8pt font_weight=bold font_style=italic cellwidth=8mm]` modifies the font size, weight, style and cell width. Readers may refer Pete Lund [1] or a complete list of ODS Style Attributes. For user's convenience, Lauren Haworth [2] attached a chart of color names at the end of her paper.

The result is shown below.

***Example 1: Lab Test - Highlight Rows and Columns with PROC REPORT / CALL DEFINE,***

PID	Visit	Visit Date	Lab Test	Value	Unit	Range Low	Range High	Abnormal Flag
1001	BASELINE	20APR2010	ALBUMIN	39	g/L	32	50	N
			ALT	35	U/L	0	55	N
			ALKALINE PHOSPHATASE	154	U/L	37	147	H
			AST	33	U/L	0	45	N
			BICARBONATE	29	mmol/L	23	30	N
			BILIRUBIN, TOTAL	8.6	umol/L	5.1	25.7	N
			CREATININE	62	umol/L	62	115	N
			SODIUM	137	mmol/L	135	148	N
	CYCLE 1 DAY 1	30APR2010	ALBUMIN	41	g/L	32	50	N
			ALT	29	U/L	0	55	N
			ALKALINE PHOSPHATASE	170	U/L	37	147	H
			AST	33	U/L	0	45	N
			BICARBONATE	28	mmol/L	23	30	N
			BILIRUBIN, TOTAL	8.6	umol/L	5.1	25.7	N
			CREATININE	71	umol/L	62	115	N
			SODIUM	134	mmol/L	135	148	L

If we just want to highlight the cells of lab value with abnormal flag equals to "HIGH" or "LOW", the COMPUTE statement on the above can be changed as below.

```

COMPUTE sihi;
  IF input(sival,best.) > sihi
    THEN CALL DEFINE('_c6_', "style", "STYLE=[BACKGROUND=RED]");
ENDCOMP;
COMPUTE silo;
  IF input(sival,best.) < silo
    THEN CALL DEFINE('_c6_', "style", "STYLE=[BACKGROUND=BLUE]");
ENDCOMP;
run;

```

The trick for COMPUTE blocks of SIHI is that SIHI has to be behind SIVAL in the COLUMN statement. Because SAS® read in the value of SIVAL first, and then compare with SIHI, so the variable has to be arranged in the right order. Otherwise, it will end up with an unexpected result.

***Example 2: Lab Test - Highlight Cells with PROC REPORT / CALL DEFINE***

PID	Visit	Visit Date	Lab Test	Value	Unit	Range Low	Range High
1001	BASELINE	20APR2010	ALBUMIN	39	g/L	32	50
			ALT	35	U/L	0	55
			ALKALINE PHOSPHATASE	154	U/L	37	147
			AST	33	U/L	0	45
			BICARBONATE	29	mmol/L	23	30
			BILIRUBIN, TOTAL	8.6	umol/L	5.1	25.7
			CREATININE	62	umol/L	62	115
			SODIUM	137	mmol/L	135	148
	CYCLE 1 DAY 1	30APR2010	ALBUMIN	41	g/L	32	50
			ALT	29	U/L	0	55
			ALKALINE PHOSPHATASE	170	U/L	37	147
			AST	33	U/L	0	45
			BICARBONATE	28	mmol/L	23	30
			BILIRUBIN, TOTAL	8.6	umol/L	5.1	25.7
			CREATININE	71	umol/L	62	115
			SODIUM	134	mmol/L	135	148

**METHOD 3: PROC REPORT PLUS PROC FORMAT**

In order to highlight CELLS according to the value of the cells, a color format can be set up to hook up the values with colors. It will be very easy to use the color format to redefine STYLE in PROC REPORT. In the following example, lab abnormal flag is highlighted based on its value.

```

proc format;
  value $ flgcolor "H"="RED" "L"="BLUE" "N"="SNOW";

```

```

run;

proc report data=db.clab nowd headline headskip spacing=1 split='~';
column pt visname labdate lbseq labtest sival siunit silo sihi labflag;
define pt / order width=5 'PID' left;
define visname / order width=14 'Visit' left;
define labdate / order width=10 'Visit Date' center;
define lbseq / order noprint;
define labtest / display width=25 'Lab Test' left;
define sival / display width=8 'Value' right;
define siunit / display width=10 'Unit' center;
define silo / display width=8 'Range~Low' right;
define sihi / display width=8 'Range~High' right;
define labflag / display width=10 'Abnormal~Flag' center
        STYLE = {BACKGROUND=$flgcolor.};
COMPUTE pt;
    CALL DEFINE(_COL_, "style", "STYLE=[BACKGROUND=MISTYROSE]");
ENDCOMP;
COMPUTE visname;
    CALL DEFINE(_COL_, "style", "STYLE=[BACKGROUND=MISTYROSE]");
ENDCOMP;
COMPUTE labdate;
    CALL DEFINE(_COL_, "style", "STYLE=[BACKGROUND=MISTYROSE]");
ENDCOMP;
COMPUTE labtest;
    CALL DEFINE(_COL_, "style", "STYLE=[BACKGROUND=LIGHTYELLOW]");
ENDCOMP;
run;

```

*Example 3: Lab Test - Highlight Cells of Abnormal flag with PROC REPORT and PROC FORMAT*

PID	Visit	Visit Date	Lab Test	Value	Unit	Range Low	Range High	Abnormal Flag
1001	BASELINE	20APR2010	ALBUMIN	39	g/L	32	50	N
			ALT	35	U/L	0	55	N
			ALKALINE PHOSPHATASE	154	U/L	37	147	H
			AST	33	U/L	0	45	N
			BICARBONATE	29	mmol/L	23	30	N
			BILIRUBIN, TOTAL	8.6	umol/L	5.1	25.7	N
			CREATININE	62	umol/L	62	115	N
			SODIUM	137	mmol/L	135	148	N
	CYCLE 1 DAY 1	30APR2010	ALBUMIN	41	g/L	32	50	N
			ALT	29	U/L	0	55	N
			ALKALINE PHOSPHATASE	170	U/L	37	147	H
			AST	33	U/L	0	45	N
			BICARBONATE	28	mmol/L	23	30	N
			BILIRUBIN, TOTAL	8.6	umol/L	5.1	25.7	N
			CREATININE	71	umol/L	62	115	N
			SODIUM	134	mmol/L	135	148	L

## METHOD 4: ODS ESCAPECHAR – MORE FLEXIBILITY

So far, we have summarized the methods of highlighting a table by specifying the STYLE=option in PROC REPORT. Another more powerful and more flexible method is to use ODS ESCAPECHAR, which allows user to specify the style anywhere in the program, not only in PROC REPORT, but also in other procedures, in the definition of titles and footnotes, or maybe in the data sets.

The ODS ESCAPECHAR statement defines a character, typically a caret (^) or tilde (~), to designate the beginning of formatting commands. For example,

```
ods escapechar = '^';
```

Then user can attach the escape sequence “^S{” to the text where the format need to be changed.

Two escape sequences “^S{” were used in the above code. The first “^S{” redefine the font size as 14 pt and background color as yellow, and the second “^S{” define the text following Title 1 as default format. S need to be capitalized in the “^S{” sequence.

This method is found to be very practical and easy to use in some complicated situation. For example, in the following table, the lab tests of a patient are listed horizontally by visit. The test value, unit, lab range, abnormal flag are all combined into one cell per test per visit. With the first three methods, it is really hard and even impossible to highlight such cells.

With the escape sequence “^S{”, it becomes a fairly easy job. When creating the data set to combine the lab value, unit, lab range, and abnormal flag, one just need to conditionally attach the sequence with proper background color definition to the combined value, according to the abnormal flag value. Without any special setting in PROC REPORT, the cell with abnormal lab value will be highlighted with the designated colors. Here below is the code to define style with the escape sequence in data step. Figure 5 shows the highlighted table.

```
val=compress(sival)||' '||compress(siunit)||' ('||compress(put(silo,best.))||', '||  
compress(put(sihi,best.))||') - '||compress(labflag);
```

```
if index(upcase(labflag),"H") then  
  val="^S={background=#ff8080}"||trim(left(val))||" ^S={}";  
else if index(upcase(labflag),"LOW") then  
  val="^S={background=#8080ff}"||trim(left(val))||" ^S={}";
```

After combining variables, embedding ESCAPECHAR ^S{ with desirable styles, and some other manipulation, the original dataset becomes the following.

labtest	vis1	vis2	vis3
^S={background=lightgrey font_weight=bold just=center}Visit Date ^S={}	^S={background=#E0E0E0}20APR2010^S={}	^S={background=#E0E0E0}30APR2010^S={}	^S={background=#E0E0E0}24JUN2010
ALBUMIN	39 g/L (32, 50) - NORMAL	41 g/L (32, 50) - NORMAL	38 g/L (32, 50) - NORMAL
ALKALINE PHOSPHATASE	^S={background=#8080ff}154 U/L (37, 147) - HIGH ^S={}	^S={background=#8080ff}170 U/L (37, 147) - HIGH ^S={}	131 U/L (37, 147) - NORMAL
ALT	35 U/L (0, 55) - NORMAL	29 U/L (0, 55) - NORMAL	24 U/L (0, 55) - NORMAL
AST	33 U/L (0, 45) - NORMAL	33 U/L (0, 45) - NORMAL	21 U/L (0, 45) - NORMAL
BICARBONATE	29 mmol/L (23, 30) - NORMAL	28 mmol/L (23, 30) - NORMAL	26 mmol/L (23, 30) - NORMAL
BILIRUBIN, TOTAL	8.6 umol/L (5.1, 25.7) - NORMAL	8.6 umol/L (5.1, 25.7) - NORMAL	5.1 umol/L (5.1, 25.7) - NORMAL
CREATININE	62 umol/L (62, 115) - NORMAL	71 umol/L (62, 115) - NORMAL	71 umol/L (62, 115) - NORMAL
SODIUM	137 mmol/L (135, 148) - NORMAL	^S={background=#8080ff}134 mmol/L (135, 148) - LOW ^S={}	137 mmol/L (135, 148) - NORMAL

With the following SAS® code, a beautiful PDF table is created, as show in Example 4.

```

%let linestr=Pt=^S={font_weight=bold}1001^S={};
%let linestr=&linestr ^S={font_weight=bold}CHEMISTRY^S={};

proc report data=final nowd missing headline headskip spacing=1 split="~"
contents="";
column id labgrp labtest vis1 vis2 vis3;
define id/noprint;
define labgrp/noprint;
define labtest/flow id style(column)={cellwidth=&idlen.% font_size=1}
style(header)={font_size=1 background=lightblue};
define vis1 / center flow style(column)={cellwidth=&vlen.% font_size=&vsize}
style(header)={font_size=&vsize background=lightblue};
define vis2 / center flow style(column)={cellwidth=&vlen.% font_size=&vsize}
style(header)={font_size=&vsize background=lightblue};
define vis3 / center flow style(column)={cellwidth=&vlen.% font_size=&vsize}
style(header)={font_size=&vsize background=lightblue};
compute before _page_ / style=[background=HONEYDEW just=left];
line "&linestr";
endcomp;
run;

```

*Example 4: Highlight with ODS ESCAPECHAR ^S{}*

Pt=1001 CHEMISTRY			
Test name	BASELINE	CYCLE 1 DAY 1	CYCLE 2 DAY 1
Visit Date	20APR2010	30APR2010	24JUN2010
ALBUMIN	39 g/L (32, 50) - NORMAL	41 g/L (32, 50) - NORMAL	38 g/L (32, 50) - NORMAL
ALKALINE PHOSPHATASE	154 U/L (57, 147) - HIGH	170 U/L (57, 147) - HIGH	131 U/L (57, 147) - NORMAL
ALT	35 U/L (0, 55) - NORMAL	29 U/L (0, 55) - NORMAL	24 U/L (0, 55) - NORMAL
AST	33 U/L (0, 45) - NORMAL	33 U/L (0, 45) - NORMAL	21 U/L (0, 45) - NORMAL
BICARBONATE	29 mmol/L (23, 30) - NORMAL	28 mmol/L (23, 30) - NORMAL	26 mmol/L (23, 30) - NORMAL
BILIRUBIN, TOTAL	8.6 umol/L (5.1, 25.7) - NORMAL	8.6 umol/L (5.1, 25.7) - NORMAL	5.1 umol/L (5.1, 25.7) - NORMAL
CREATININE	62 umol/L (62, 115) - NORMAL	71 umol/L (62, 115) - NORMAL	71 umol/L (62, 115) - NORMAL
SODIUM	137 mmol/L (135, 148) - NORMAL	134 mmol/L (135, 148) - LOW	137 mmol/L (135, 148) - NORMAL

## CONCLUSION

Four methods to highlight PDF output have been presented in this paper. These methods can also be used to modify other attributes. Each method has its own advantages and disadvantages. You may select an appropriate method according to your special needs and programming habits. Some simple examples are given to illustrate the basic ideas of the methods. You may refer to the reference behind to get more detailed information and further explore the powerful tools provided by ODS.

## REFERENCE

[1] Lund, Pete, "PDF Can be Pretty Darn Fancy – Tips and Tricks for the ODS PDF Destination", SUGI 31, Paper 092-31.

[2] Haworth, Lauren. 2005. "SAS® with Style: Creating your own ODS Style Template for PDF Output.", SUGI 30, Paper 132-30.

<http://www2.sas.com/proceedings/sugi30/132-30.pdf>

[3] Boberg, Wendy, "PROC REPORT in Color ... What's Your STYLE?", SAS Global Forum 2008, Paper 224-2008.

## ACKNOWLEDGMENTS:

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

## CONTACT INFORMATION

**Xuefeng Yu**  
**Celgene Corporation**  
**86 Morris Ave**  
**Summit, NJ 07901-3915**  
**Work Phone: (908) 860-7438**  
**Fax: (908) 766-8798**  
**Email: xuyu@celgene.com**